Implementing a programming language with a dependent type system

Eduardo Henke¹

¹ Departamento de Informática e Estatística – Universidade Federal de Santa Catarina (UFSC)

eduardo.henke@grad.ufsc.br

Abstract. The main goal of this project is to explore the potential benefits of advanced type systems, in software development, through an implementation of a dependently typed programming language. This could help improve the reliability and safety of software by enabling static verification of arbitrary properties about the code. This work consists in an extension of the lambda calculus with dependent types, which allows us to write programs that not only have the ability to perform computations, but whose correctness can also be proven.

1. Introduction

Software development is a big area, which it is being progressively valued over time. However, due to the complexity of maintaining and developing large software systems, bugs are frequent, where there is a discrepancy in what the developer expected to happen, with what was written in the software. Because of this, we can see the importance of being able to guarantee that the software works as we expect, and it is also for this reason that the semantic analysis part of a language is used, where we can verify if the behavior expected by the developer reflects what was in fact written.

In the semantic analysis phase, the compiler checks if the code written by the developer is in accordance with the problem modeling. One of the most basic mechanisms for this is *types*, which we use to define which variables are of a given set of possible values. With a simple type system [Pierce 2002], we can verify that a variable of type *string* cannot receive a value of type *int*.

With a more advanced type system, i.e. with dependent types [Pierce 2004] we can:

- specify invariants (rules that must hold) that will be statically checked in the code, for example:
 - in a banking system, during a withdrawal operation, the amount withdrawn cannot be greater than the balance from a bank account¹:
 - -- given the account balance, an amount and a proof
 - -- that the withdrawn amount is less than the balance

 $^{^{1}}$ We use a Haskell-style pseudo-code notation to describe this code, explained in more details in subsection 3.2 and in subsection 2.3

```
-- perform the operation withdraw_from : (\texttt{account} : \texttt{Account}) \to \\ (\texttt{amount} : \texttt{Nat}) \to \\ (\texttt{amount} <= \texttt{account.balance}) \to \\ \texttt{Nat}
```

- in most programs, we can have a list of elements, and we can have a function that returns the first element of the list, but what happens if the list is empty? We can use dependent types to specify that the list cannot be empty²:

```
head:
         -- given any type A
         (A : Type) \rightarrow
         -- given any number
         (n : Nat) \rightarrow
         -- given a vector of type A, with length n+1
         -- (note that this means that even if n is 0,
         --- succ n, will be 1, and the vector will have
         -- at least one element, our language
         -- STATICALLY INVALIDATES all uses of head on an empty
             list )
         Vec A (succ n) \rightarrow
         -- return the first element of the vector
         Α
-- returns the length of the resulting vector as a type
\texttt{append} \; : \; (\texttt{A} \; : \; \texttt{Type}) \; \; \rightarrow \; (\texttt{n} \; : \; \texttt{Nat}) \; \; \rightarrow \; (\texttt{m} \; : \; \texttt{Nat}) \; \; \rightarrow \; \texttt{Vec} \; \; \texttt{A} \; \; \texttt{m}

ightarrow Vec A n 
ightarrow Vec A (plus m n)
append = ... -- implementation is ommitted
```

• prove properties (and theorems) about the code, because we can encode logical propositions as types [nLab authors 2022d], and their respective proofs as code (evidence of that type) [nLab authors 2022c], e.g. prove that an operation inserting an element into an ordered list does not change the order of the list, or prove that a mathematical relation is associative:

```
-- proof that addition on Naturals is associative -- given any three numbers (m, n, p), we can prove that: -- (m + n) + p = m + (n + p) plus_assoc : (m n p: Nat) \rightarrow ((plus (plus m n) p) = (plus m (plus n p))) plus_assoc = -- ... proof is ommitted
```

²In this case, the first two parameters are being passed explicitly, which can be cumbersome to the developer's experience, that is why most dependently-typed languages have some *inference* mechanisms, which allows the compiler to infer some parameters being passed, like the element type and length of the vector

1.1. Existing work

Some of the most well-known dependently-typed languages are Agda [Mörtberg and Norell 2022], Idris [Team 2022b], Coq [Team 2022a] and Lean [Team 2022c], all of which allow for static verification and the ability to prove arbitrary properties about the code. These languages are based on different underlying type theories, which shape the foundations of their type systems and the capacity to specify and prove theorems. Additionally, the focus of these languages can vary; for instance, Agda is more geared towards programming, while Coq has a stronger emphasis on theorem proving.

1.2. Objectives

The general objective of this thesis is to examine the potential advantages of utilizing advanced type systems, such as dependent types, in software development, through the implementation of a dependently-typed programming language. We will provide a deeper understanding of how such languages work behind the scenes, and how they can be used to statically verify properties about a program.

To support this general objective, we will pursue the following specific objectives:

- Investigate existing work on dependently-typed programming languages and advanced type systems
- Develop a dependently-typed programming language as a proof-of-concept implementation. As the focus is on the type system, we will present only the type-checker of such language.
- Use the language to encode business invariants and demonstrate the ability to prove properties about the code, including through the use of static type checking and automated theorem proving.

We believe that achieving these objectives will provide valuable insights into the potential of using advanced type systems in software development.

2. Theoretical Basis

In this section, we are going to provide some background in lambda calculus, inference rules and some type theories.

2.1. Lambda Calculus

When we do not have the tool of abstraction, calculations such as the following seem complex to follow:

$$(2*1) + (7*6*5*4*3*2*1) + (5*4*3*2*1)$$

That is why we can abstract the underlying common concepts and define a function to capture that common abstraction, as defined here:

```
factorial = \lambda n . if n == 0 then 1 else n * (factorial (n - 1))
```

Now we can use the previously defined function³ to more concisely define the previously cumbersome calculation, as shown here:

```
(factorial 2) + (factorial 7) + (factorial 5)
```

We have captured an essential understanding of the calculation, and abstracted it into a concept that we can later reuse.

Lambda calculus captures the essential mechanisms of a programming language based on a few simple rules (abstraction, application). It was invented by Alonzo Church in the 1920s [Pierce 2002] and it is a *logical calculus* or *formal system*

Definition 2.1 (Term). A *term* is an expression in a given formal language that represents a value or a computation.

The terms of the lambda calculus are generated by the following grammar:

```
t := x variable | \lambda x.t  abstraction | t t  application
```

Figure 1. Lambda calculus grammar

A few example terms of the above grammar:

- x, a simple (unbounded) variable
- $\lambda x.x$, the identity function
- $\lambda x.\lambda y.x$, a function receiving argument x, that returns another function receiving argument y, that returns x
- $(\lambda x.x)$ y, applying y to the identity function

2.1.1. Evaluation rules

This calculus primary benefit is in its evaluation, or computation. For example, let's take the above mentioned term and evaluate it: $(\lambda x.x)$ y in one step evaluates to y, because we are applying the identity function to the variable y. But how does that work for all terms? How can we formalize it?

Definition 2.2 (Evaluation). $a \rightarrow b$, a term a can evaluate to term b

For that we need inference rules. Inference rules are syntactic transformation rules, i.e. they only need to check the form of the term and with that form they transform the original term to something else. We denote the inference rules by having the premises⁴ above the bar and the conclusions/consequent below it, like the following inference rule symbolizing the property that adding zero to a number does not change the number:

³This concept will be later formalized as "abstraction"

⁴When we do not need any pre-conditions we simply write nothing above the bar

$$\frac{a\in\mathbb{N}}{a+0\to a}(\mathrm{Add\text{-}Zero})$$

Definition 2.3 (Value). For a term to be a value, it must have the form of a function, i.e. only terms that are an abstraction ($\lambda x.t$) are considered to be values. For ease of notation, we will use t to denote any term, and v (and its derivatives) to denote values e.g. v_1 .

Definition 2.4 (Substitution). The process of substituting a variable x with another term y in a given term t is denoted by $[x \mapsto y]t$

We will be denoting the evaluation of a lambda calculus terms using a system of inference rules written below:

$$\frac{\mathtt{t_1} \to \mathtt{t_1'}}{\mathtt{t_1} \ \mathtt{t_2} \to \mathtt{t_1'} \ \mathtt{t_2}} (\text{E-App1})$$

Given a term t_1 t_2 (applying the argument t_2 to the function t_1), if the term t_1 can evaluate to another term t_1' , we can rewrite the original term to t_1' t_2 .

$$\frac{\mathtt{t_2} \to \mathtt{t_2'}}{\mathtt{v_1} \ \mathtt{t_2} \to \mathtt{v_1} \ \mathtt{t_2'}} (\text{E-App2})$$

Given a term v_1 t_2 (applying the argument t_2 to a term which is a value v_1), if the term t_2 can evaluate to another term t_2' , we can rewrite the original term to v_1 t_2' .

$$\frac{1}{\lambda \mathtt{x.t_{12}} \ \mathtt{v_2} \rightarrow [\mathtt{x} \mapsto \mathtt{v_2}] \mathtt{t_{12}}} (\text{E-AppAbs})$$

Whenever we have an application and the argument is a value, we can rewrite the original term to the body of the function, while replacing the parameter (x) to the argument (v_2) .⁵⁶

Definition 2.5 (Normal form). When we have no more evaluation steps to perform, i.e. none of the above rules can be applied, and the resulting term is a *value* we say that the term is in *normal form*. For example the term $\lambda x.x$ is in normal form.

At first it does not seem that this simple system can compute the same terms that an ordinary programming language can, but it turns out we can encode numbers, data structures (lists, sets, etc.), *if* expressions [Pierce 2002]. It was proved that this model is equivalent to a Turing Machine [Church 1941].

⁵This rule is also called a beta reduction [Pierce 2002].

⁶Note that we do not have any preconditions for this rule, whenever we *syntactically* have an application whose argument is a value, we can reduce it.

2.2. Simply-Typed Lambda Calculus

Up to now, we have only had the notion of evaluation of terms, we are able to define computation steps and an engine can run those for us. But without much insight into what "types" of terms we are writing we can easily make the underlying evaluation engine try to compute a program that never halts [Pierce 2002], or if we add to the lambda calculus' rules operations that can only be applied to numbers and feed them with booleans that can also make the engine stuck, i.e. passing a program that cannot be computed.

Those problems can be solved if we somehow inspect the original program before evaluating it, and if desired properties can be derived from only the program specification, checking those properties against the code. One way to do it is by the concept of *types* [Pierce 2002].

Types allow us to define a set of possible values a term may have during runtime. When we talk about a variable having a type of Nat (Natural numbers set), it is telling that during runtime that variable can only possess the values 0, 1, 2, When we define a variable of type Bool, the only possible values in runtime are True, False.

We can also annotate various parts of our program that allows this checker (which we will call typechecker from now), to verify the correctness of our program.

2.3. Dependently-Typed Lambda Calculus

The simply-typed system provides us some basic tools to define types, but we need to be able to define types for more complex terms.

Dependent types allow types to depend on the terms themselves, i.e. we do not have this stark distinction of types and terms. That allows us greater freedom in specifying types, which are a foundation on which our typechecker can verify the correctness of our code [Pierce 2004].

Like many dependently-typed languages, we will show the typing rules (Figure 3) and a grammar (Figure 2) with the same *syntax* for the terms and types, however for clarity we will be using lowercase letters for terms and uppercase letters for their types.

Figure 2. Dependently-typed lambda calculus grammar

$$\frac{x:T\in\Gamma}{\Gamma\vdash x:T}(\text{T-Var})$$

$$\frac{\Gamma,x:T_1\vdash y:T_2\qquad\Gamma\vdash T_1:\mathbf{Type}}{\Gamma\vdash \lambda x.y:(x:T_1)\to T_2}(\text{T-Lambda})$$

$$\frac{\Gamma\vdash t_1:(x:T_{11})\to T_{12}\qquad\Gamma\vdash t_2:T_{11}}{\Gamma\vdash t_1:t_2:[x\mapsto t_2]T_{12}}(\text{T-App})$$

$$\frac{\Gamma\vdash T_1:\mathbf{Type}\qquad\Gamma,x:T_1\vdash T_2:\mathbf{Type}}{\Gamma\vdash (x:T_1)\to T_2:\mathbf{Type}}(\text{T-Pi})$$

$$\frac{\Gamma\vdash \mathbf{Type}:\mathbf{Type}}{\Gamma\vdash (x:T_1)\to T_2:\mathbf{Type}}(\text{T-Type})$$

Figure 3. Dependently-typed lambda calculus typing rules

Definition 2.6 (Context). A typing context, denoted by Γ, is a sequence of variables and their associated types. The empty context is written as \emptyset . We can extend an existing context Γ by adding a new variable with its associated type to it: Γ, x: T.⁷

The existing typing rule will change from a two-place relation $\mathbf{x}:T$ to a three-place relation $\Gamma \vdash \mathbf{x}:T$, meaning that \mathbf{x} has type T under the context Γ , which will provide assumptions to the types of the free variables⁸ in \mathbf{x} .

Also, where before we only had that the type of a function is $T_1 \to T_2$, now we have that the type of a function⁹ is $(x:T_1) \to T_2$. That means that the function's return type expression T_2 can depend on the function's argument x, e.g. the type of a function that inserts a Bool in a length-indexed vector of Bools can be $(n:Nat) \to Bool \to Vec Bool (n+1)$. The rules are explained as follows:

- Rule T Var, to check if a variable x has type T in the context Γ , we check if x : T is in Γ .
- Rule T-Lambda, a lambda expression $\lambda x.t_2$ has type $(x:T_1) \to T_2$ in context Γ , if when we extend the context Γ to include the variable x with type T_1 , we can check if t_2 has type T_2 in the extended context, and also if T_1 is indeed a type.
- Rule T-App, a function application t_1 t_2 has type T_{12} (with the x variable substituted by the t_2 parameter) in context Γ , if we can check that t_1 has type $(x:T_{11}) \to T_{12}$ in the context Γ , and if we can check that t_2 has type T_{11} in the context Γ .
- Rule T Pi, a Pi type $(x : T_1) \to T_2$ has type **Type** in context Γ , only if T_1 also has type **Type** in context Γ , and if T_2 has type **Type** in the extended context Γ , $x : T_1$.
- Rule T Type, the term Type also has type Type.

We will add two extensions to our grammar to aid in creating programs:

⁷We will assume that each of the variables in this list are distinct from each other, so that there will always be at most one assumption about any variable's type.

⁸Variables that are not bound to any lambda binder

⁹Formally called a *Pi* type

```
\begin{array}{cccc} \textbf{t}, \textbf{T} & ::= ... \\ & & | \textbf{t} : \textbf{T} & & \text{type annotation} \\ & & | \texttt{name} = \textbf{t} & \text{assigning a name to a term} \end{array}
```

Figure 4. Dependently-typed lambda calculus syntax sugar

Which respectively mean:

- An expression can be annotated with a type, e.g. x : Nat, and that will trigger the typechecker to check that $\Gamma \vdash x : Nat$ given the underlying context.
- A name can be assigned to a term, e.g. $id = \lambda x.x$.

With that we can write this polymorphic identity function program annotated with its type and its associated typing proof tree:

$$id: (x: \mathbf{Type}) \rightarrow (y:x) \rightarrow x$$

 $id = \lambda x. \lambda y. y$

We can derive a proof tree proving that the type of id is in fact what was annotated:

$$\frac{\Gamma, x: \mathbf{Type} \vdash \frac{\Gamma, y: x \vdash y: x}{\Gamma, y: x \vdash y: x} \frac{\frac{x: \mathbf{Type} \in \Gamma}{x: \mathbf{Type}} (\text{T-Var})}{\lambda y. y: (y: x) \to x} \frac{\Gamma, x: \mathbf{Type} \in \Gamma}{\Gamma \vdash \lambda x. \lambda y. y: (x: \mathbf{Type}) \to (y: x) \to x} \frac{\Gamma \vdash \lambda x. \lambda y. y: (x: \mathbf{Type}) \to (y: x) \to x}{\Gamma \vdash \lambda x. \lambda y. y: (x: \mathbf{Type}) \to (y: x) \to x} (\text{T-Lambda})$$

The goal of this project is to derive the typing proof tree automatically.

2.3.1. Definitional Type Equality

Definition 2.7 (Judgement). A judgement is a proposition that is made on a given term. The previously defined *typing rule* is a form of judgement.[nLab authors 2022b]

Definition 2.8 (Definitional equality). Definitional equality is a judgement of the form: $\Gamma \vdash A = B$. Classically, definitional equality is called intensional equality¹⁰. However in this thesis we will define it to mean both intensional **and** computational equality¹¹ [nLab authors 2022a].

This judgement is defined by the properties stated in Figure 5. Rule E-Beta ensures that beta-equivalence is contained in this judgement, because terms that evaluate to each other should be equal. Rules E-Refl, E-Sym, and E-Trans allows this judgement to be considered an equivalence relation [Weirich 2022].

¹⁰Intensional equality is the relation generated by abbreviatory definitions, changes of bound variables and the principle of substituting equals for equals [nLab authors 2022a]

¹¹Computational equality is the relation generated by various reduction rules, e.g. beta reduction [nLab authors 2022a]

$$\overline{\Gamma \vdash (\lambda x.a) \ b} = [x \mapsto b]a^{(E-Beta)}$$

$$\overline{\Gamma \vdash A} = A^{(E-Refl)}$$

$$\overline{\Gamma \vdash A} = B \text{(E-Sym)}$$

$$\underline{\Gamma \vdash A_1 = A_2 \qquad \Gamma \vdash A_2 = A_3} \text{(E-Trans)}$$

$$\underline{\Gamma \vdash A_1 = A_2 \qquad \Gamma, x : A_1 \vdash B_1 = B_2} \text{(E-Pi)}$$

$$\overline{\Gamma \vdash (x : A_1) \rightarrow B_1 : (x : A_2) \rightarrow B_2} \text{(E-Pi)}$$

$$\underline{\Gamma, x : A_1 \vdash b_1 = b_2} \text{(E-Lam)}$$

$$\underline{\Gamma \vdash a_1 = a_2 \qquad \Gamma \vdash b_1 = b_2} \text{(E-App)}$$

$$\underline{\Gamma \vdash a_1 = a_2 \qquad \Gamma \vdash a_1 = a_2} \text{(E-App)}$$

$$\underline{\Gamma, x : A \vdash b : B \qquad \Gamma \vdash a_1 = a_2} \text{(E-Lift)}$$

$$\underline{\Gamma \vdash (x \mapsto a_1]b = [x \mapsto a_2]b} \text{(E-Lift)}$$

$$\underline{\Gamma \vdash (a_1 : A) = a_2} \text{(E-Annot)}$$

Figure 5. Inference rules for the definitional type equality judgement

3. Implementation

The rules presented on subsection 2.3 were developed to allow typing proof trees to be built, however they were not developed having in mind how these proof trees can be built, i.e. these rules are not syntax-directed, we can not devise a decidable algorithm based on these rules alone. The reason for that is because rule T-Lambda is not syntax-directed, it is not clear what is the type of the argument of the function (we extend the context with it, because it is necessary to typecheck the body of the function). Because of that we need to revise the existing rules and transform them into syntax-directed rules.

3.1. Bidirectional type system

One way to automatically derive proof trees for a given program is to define a bidirectional type system, which is based on two types of judgements:

• Type inference, denoted as $\Gamma \vdash \mathbf{x} \Rightarrow \mathbf{T}$ which given a term \mathbf{x} and a context Γ , will infer(return) the type of that term.

```
inferType :: Context 
ightarrow Term 
ightarrow Maybe Type
```

• Type checking, denoted as $\Gamma \vdash x \Leftarrow T$ which given a term x, a context Γ and a type T, will check that the term is of the given type.

$$\texttt{checkType} \; :: \; \texttt{Context} \; \to \texttt{Term} \; \to \texttt{Type} \; \to \texttt{Bool}$$

We now can develop inference rules that represent the algorithmically-feasible version of the subsection 2.3 rules. Both types of judgements will be used, the type inference rules can use the type checking rules, and vice versa.

$$\begin{split} \frac{x: A \in \Gamma}{\Gamma \vdash x \Rightarrow A} (\text{I-Var}) \\ \frac{\Gamma \vdash a \Rightarrow (x: A) \to B \qquad \Gamma \vdash b \Leftarrow A}{\Gamma \vdash a \Rightarrow [x \mapsto b]B} (\text{I-App-Simple}) \\ \frac{\Gamma \vdash A \Leftarrow \mathbf{Type} \qquad \Gamma, x: A \vdash B \Leftarrow \mathbf{Type}}{\Gamma \vdash (x: A) \to B \Rightarrow \mathbf{Type}} (\text{I-Pi}) \\ \frac{\Gamma \vdash \mathbf{Type} \Rightarrow \mathbf{Type}}{\Gamma \vdash (a: A) \Rightarrow A} (\text{I-Annot}) \end{split}$$

Figure 6. Bidirectional type-inference rules

$$\begin{split} \frac{\Gamma, x : A \vdash a \Leftarrow B & \Gamma \vdash A \Leftarrow \mathbf{Type}}{\Gamma \vdash \lambda x. a \Leftarrow (x : A) \to B} (\text{C-Lambda}) \\ & \frac{\Gamma \vdash a \Rightarrow A}{\Gamma \vdash a \Leftarrow A} (\text{C-Infer-Simple}) \end{split}$$

Figure 7. Bidirectional type-checking rules

3.2. Modules

We will also introduce the concept of modules and top-level declarations to aid in the development of programs. Where a module consists of a list of declarations that are defined once and used throughout the module:

$$\begin{array}{ll} \texttt{Decl} ::= \texttt{t} : \texttt{T} & \texttt{type-signature} \\ & | \texttt{t} = \texttt{t} & \texttt{definition} \end{array}$$

Figure 8. Module grammar

Here are some examples:

3.2.1. Type-checking

To type-check a module we use the checkType function when we have an associated type-signature for a given name, and we use the inferType function when we do not have a type-signature for a given name. Also we put the existing declarations in the scope of all subsequent type-checks.

3.3. Data Types

We will also introduce the concept of data types, to aid the programmer in having structured data in their program:

Figure 9. Data type grammar

Definition 3.1 (Telescope). A telescope is a list of declarations, it is called like that because of the scoping behavior of this structure [Weirich 2022]. The scope of each variable merges with all of the subsequent ones, e.g.:

```
-- notice the scoping behavior of this structure
(A : Type) (n : Nat) (v : Vec A n)
-- here we are showing that the telescope can use both
-- a type annotation, and a definition, which symbolizes
-- a constraint being put upon any variable, in this
-- case, it requires n to be of value Zero
(n : Nat) [n = Zero]
```

Here we are showing an example, by first defining some data types, then using it as terms:

```
data Bool : Type where {
    False, -- False is a case from Bool, it is a data type constructor
    True -- as well as True
} -- Bool is a data definition with an empty telescope

data Nat : Type where {
    Zero,
```

```
-- Succ is a data type constructor with
-- a Telescope of one argument (another Nat number)
Succ of (Nat)
}

data List (A : Type) : Type where {
    Nil,
    -- Cons is a data type constructor with
    -- a Telescope of two arguments (element of A and a List of A)
    Cons of (A) (List A)
}

-- t is annotated with a type, constructed by applying a
-- list of arguments (Bool) to the type constructor (List)
t : List Bool
-- t is defined using a data type constructor (Cons) by applying
-- a list of arguments (True, Nil) to it
t = Cons True Nil
```

Next, we are showing that the telescope can have a constraint on the previous values, with a Vector data type:

```
data Vec (A : Type) (n : Nat) : Type where {
 -- Nil is a data type constructor with
 -- a constraint that n (provided by the type) must be zero
 Nil of [n = Zero],
 -- Cons is a data type constructor with a Telescope of
 -- three arguments (Nat m, element of A, and a Vec of A with length m)
 -- and one constraint that the Vector built by Cons,
 -- must have length m+1, m being the underlying vector
 -- which this Cons was built upon
 Cons of (m : Nat) (A) (Vec A m) [n = Succ m]
-- t is annotated with a type, constructed by applying a
-- list of arguments (Bool) to the type constructor (List)
t : Vec Bool 2
-- t is defined using a data type constructor (Cons) by applying
-- a list of arguments (True, Nil) to it
t = Cons 1 True (Cons 0 False Nil)
```

We type-check type and data-type constructor applications according to the telescope and the arguments provided, e.g.:

```
-- compares the telescope (A : Type) (n : Nat)
-- with the arguments provided: Bool, 1
-- Bool is of type Type, and 1 is of type Nat
v : Vec Bool 1
-- compares the telescope (m : Nat) (A : Type) (v : Vec A m)
-- with the arguments provided: 0, True, Nil
```

```
-- 0 is of type Nat, True is of type A (Bool), and Nil is of type Vec Bool 0  v = {\rm Cons} \ 0 \ {\rm True} \ {\rm Nil}
```

3.3.1. Pattern-matching

```
\texttt{t}, \texttt{T} ::= ... \\ | \textit{case} \; \texttt{t} \; \textit{of} \; \texttt{Case} * \quad \text{pattern matching of a term} \\ \\ \texttt{Case} ::= \texttt{Pattern} \; \rightarrow \; \texttt{t} \\ \\ \texttt{Pattern} ::= \texttt{DConName} \; \texttt{PatVar} * \\ \\ | \; \texttt{PatVar} \\ \\ \texttt{PatVar} ::= \texttt{identifier} \\ \\ \end{aligned}
```

Figure 10. Data type with pattern matching grammar

Here are some examples of pattern matching:

```
not : Bool \rightarrow Bool not = \lambda b. case b of { False \rightarrow True, True \rightarrow False } plus : Nat \rightarrow Nat \rightarrow Nat plus = \lambda a. \lambda b. case a of { Zero \rightarrow b, Succ a' \rightarrow Succ (plus a' b) }
```

To type-check pattern matching cases, we have to:

extend the typing context with the pattern variables, e.g. when we see Succ n' -> body, we should extend the typing context with n': Nat when we type-check the body:

```
n : Nat
case n of {
   -- we should know that n' has type Nat,
   -- according to Nat's data definition
   Succ n' → plus n' n',
   ...
}
```

• all cases must conform to the same type, e.g.:

```
\label{eq:nat_bool} \begin{array}{ll} \texttt{nat\_to\_bool} & \texttt{Nat} & \to \texttt{Bool} \\ \texttt{nat\_to\_bool} & = \lambda \texttt{n.} & \texttt{case} & \texttt{n} & \texttt{of} \end{array} \{
```

```
-- the type of the body of this case is Bool,
-- so the type of the whole case is Bool
Zero → False,
-- the same thing with this case
Succ n' → True
}
-- invalid code would have been:
-- case n of { Zero → False, Succ n' → n' }
```

• check for exhaustiveness, i.e. we should check that there is no other possible case, otherwise, we should throw an error

```
n : Nat
-- this case is exhaustive, because it covers all possible cases
case n of {
   Zero → False,
   Succ n' → True
}
-- this case is not exhaustive, because it does not cover Zero
-- case n of {
   Succ n' → True
-- }
```

• unify the scrutinee's type (term being matched) with the type of the pattern, e.g. if we are applying a case expression to a n which is a Nat, and in the Zero -> body case, we should replace the type of a from Nat to Zero, in this branch.

3.4. Project's Code

The project was developed in *Haskell* using the *Unbound* library for variable substitution (beta reduction), and was heavily inspired by the pi-forall [Weirich 2022] language.

3.4.1. Terms

The syntax of the terms, module and declarations are represented as a data type in *Haskell*. Here we have the data type for the terms, which is a recursive data type, as well as the data type for the types, which is simply an alias to the term data type:

```
module Syntax where

type TName = Unbound.Name Term -- Term names for our AST

type TCName = String -- type constructor names

type DCName = String -- data constructor names

-- because types and terms are the same in dependent typing,

-- we will alias them

type Type = Term
```

```
10
11
   data Term
12
          = Type -- type of types
13
          | Var TName -- variables: x
14
          | Lam (Unbound.Bind TName Term) -- abstractions: λx.a
15
          | App Term Term -- application: f x
          | Pi Type (Unbound.Bind TName Type) -- fn types: (x : A) \rightarrow B
16
          | Ann Term Type -- "ascription" or "annotated terms": (a: A)
17
          -- Data-type related terms
18
19
          | DCon DCName [Term] -- Just True
20
          | TCon TCName [Term] -- Maybe Bool
21
          | Case Term [Match] -- case analysis `case a of matches`
22
          -- Proof related terms
23
          | TyEq Type Type -- equality type: (plus 0 n) = n
24
          | Refl -- equality evidence: refl is of type x = x
25
          | Subst Term Term
```

Next, we have the data type for the modules and top-level declarations:

```
newtype Module = Module {declarations :: [Decl]}

-- a "top-level definition" of a module
data Decl
= TypeSig TName Type -- a : A
| Def TName Term -- a = b
| DataDef TCName Telescope [ConstructorDef] -- data Bool ...
```

And finally, we have the data type for data types and pattern-matching constructs:

```
-- a data constructor has a name and a telescope of arguments
2
   data ConstructorDef = ConstructorDef DCName Telescope
3
         deriving (Unbound.Alpha, Unbound.Subst Term)
4
5
   newtype Telescope = Telescope [Decl]
6
         deriving (Unbound.Alpha, Unbound.Subst Term)
7
   -- represents a case alternative
9
   newtype Match = Match (Unbound.Bind Pattern Term)
10
         deriving anyclass (Unbound.Alpha, Unbound.Subst Term)
11
12
   data Pattern
         = PatCon DCName [Pattern]
13
14
          | PatVar TName
15
          deriving (Eq, Generic, Unbound.Alpha, Unbound.Subst Term)
```

3.4.2. Type-checking

Instead of using a Maybe or Either type to represent the result of type checking, we will use a monad to encapsulate the behaviour of:

- failing to type-check and returning an error message
- getting and setting the state of the typing context
- generating fresh variable names

```
module Environment where
 2
 3
    import Control.Monad
    import Control.Monad.Except (ExceptT)
 4
    import Control.Monad.Reader (ReaderT)
    import qualified Unbound. Generics. Locally Nameless as Unbound
 6
 7
    type TcMonad = Unbound.FreshMT (ReaderT Env (ExceptT Err IO))
 9
10
   data Env = Env {ctx :: [Decl]}
11
12
    emptyEnv :: Env
    emptyEnv = Env {ctx = []}
13
14
15
    \mathtt{extendCtx} :: \mathtt{Decl} \ 	o \mathtt{TcMonad} \ \mathtt{a} \ 	o \mathtt{TcMonad} \ \mathtt{a}
16
    \texttt{extendCtxs} \; :: \; \texttt{[Decl]} \; \to \texttt{TcMonad} \; \texttt{a} \; \to \texttt{TcMonad} \; \texttt{a}
17 | lookupTyMaybe :: TName \rightarrow TcMonad (Maybe Type)
    {	t lookupTy} :: {	t TName} 
ightarrow {	t TcMonad} 
ightarrow {	t Type}
18
    -- implementation ...
19
```

We now have some helper functions that can be used to implement the type-checking rules, e.g. given a name of a variable we can use lookupTy to get its type, if it fails to find that variable in the context, it will stop the type-checking process and issue an error.

Also the two judgements' signatures will change to:

```
1 -- given a term return its type (or an error message)
2 inferType :: Term \rightarrow TcMonad Type
3 -- given a term and a type, check if the term is of the given type
4 -- if it is, return (), otherwise an error message
5 checkType :: Term \rightarrow Type \rightarrow TcMonad ()
```

The rules for propositional and definitional equality of terms are defined in the equate function, which given two terms will check if they are equal or not, if they are not it will throw an error.

```
equate :: Term \to Term \to TcMonad ()

two terms are equal, if they are alpha-equivalent

i.e., by just properly renaming the variables

they are the same term
```

```
equate t1 t2 | aeq t1 t2 = return ()
 6
    equate t1 t2 = do
 7
      \texttt{nf1} \; \leftarrow \texttt{whnf} \; \texttt{t1}
 8
      nf2 \leftarrow whnf t2
9
      case (nf1, nf2) of
10
        (Lam bnd1, Lam bnd2) \rightarrow do
           -- get the body of each lambda
11
12
           (_, t1, _, t2) \leftarrow unbind2Plus bnd1 bnd2
13
           -- lambdas are equal, if their bodies are equal
14
          equate t1 t2
15
        (App f1 x1, App f2 x2) \rightarrow do
16
          equate f1 f2
17
          equate x1 x2
18
        ... rest of the terms ommitted for brevity ...
19
        (\_, \_) \rightarrow err ["Expected", show nf2, "but found", show nf1]
```

- From line 10 to 14, we are checking that a lambda is equal to another lambda if their bodies are equal, according to rule E-Lam in Figure 5
- From line 15 to 17, we are checking that an application is equal to another application if their function and argument are equal, according to rule E-App in Figure 5

Definition 3.2 (Weak-head normal form). A term is said to be in weak-head normal form, when we apply weak-head normalization, which is when the leftmost, outer-most reducible expression is always selected for beta-reduction, and the process is halted as soon as the term begins with something other than a lambda abstraction. [Pierce 2004]

An unformal idea of weak-head normalization, is that it is a subset of the normal beta-reduction (computation) rules, so that we compute just enough to observe the structure of the term, e.g. when I do not care about the result of factorial 100, but I do care about its structure, if it is a Nat or not.

As shown in the above code section, we also need a function to calculate the weak-head normal form of a given term, which is defined as follows in the whnf function:

```
whnf :: Term \rightarrow TcMonad Term
 2
    -- if we are whnf-inf a variable,
 3
   whnf (Var x) = do
 4
      -- look up its definition in the context
      \texttt{maybeTm} \ \leftarrow \texttt{lookupDefMaybe} \ \texttt{x}
 5
 6
      case maybeTm of
        -- and whnf the definition
        (Just tm) \rightarrow whnf tm
 8
        -- or if there is no definition in the context
9
10
        -- return the variable
11
         \rightarrow pure (Var x)
    -- if we are whnf-ing an application,
```

```
13 \mid \text{whnf (App a b)} = \text{do}
14
      v \leftarrow whnf a
15
      -- check the type of the whnf'd function
16
      case v of
17
        -- if it is indeed a lambda abstraction
18
        (Lam bnd) \rightarrow do
19
          (x, a') \leftarrow unbind bnd
20
          -- substitute the argument for the bound variable
21
          whnf (subst x b a')
        -- otherwise, return the application
22
23
          \rightarrow return (App v b)
24
       ... rest of the terms ommited for brevity ...
25
    whnf tm = return tm
```

Next we will implement the type-checking rules for the terms and types of our language, which are defined in Figure 6.

We will use a tcTerm function that will be used to centralize the two judgements (inferType and checkType). If the second parameter is Nothing, it will enter into *inference* mode, otherwise it will enter into *type-checking* mode.

Here we will just define the inferType and the checkType function:

```
module Typechecker where
 1
 2
 3
   inferType :: Term \rightarrow TcMonad Type
   inferType t = tcTerm t Nothing
 4
 5
 6
   \texttt{checkType} \; :: \; \texttt{Term} \; \to \texttt{Type} \; \to \texttt{TcMonad} \; \; ()
    checkType tm ty = do
7
8
      -- Whenever we call checkType we should call it
9
      -- with a term that has already been reduced to
      -- normal form. This will allow rule c-lam to
10
      -- match against a literal function type.
11
12
      nf \leftarrow whnf ty
13
      ty' ← tcTerm tm (Just nf)
14
   -- Make sure that the term is a type (i.e. has type 'Type')
15
   | 	exttt{tcType} :: 	exttt{Term} 
ightarrow 	exttt{TcMonad} ()
16
   tcType tm = void (checkType tm Type)
17
```

Next, we will define the tcTerm function, which will be used to implement the type-checking rules:

```
1 \mid \texttt{tcTerm} :: \mathsf{Term} \rightarrow \mathsf{Maybe} \; \mathsf{Type} \rightarrow \mathsf{TcMonad} \; \mathsf{Type}
2 -- Infer-mode
3 tcTerm (Var x) Nothing = lookupTy x
4 tcTerm (Ann tm ty) Nothing = do
5
     checkType tm ty
6
     return ty
7
   tcTerm (Pi tyA bnd) Nothing = do
      (x, tyB) \leftarrow unbind bnd
8
9
     tcType tyA
10
      extendCtx (TypeSig x tyA) (tcType tyB)
11
     return Type
12
   tcTerm (App t1 t2) Nothing = do
      ty1 \leftarrow inferType t1
13
14
      let ensurePi :: Type \rightarrow TcMonad (TName, Type, Type)
          ensurePi (Ann a _) = ensurePi a
15
16
          ensurePi (Pi tyA bnd) = do
17
            (x, tyB) \leftarrow unbind bnd
18
            return (x, tyA, tyB)
19
          ensurePi ty = err ["Expected a function type, but found ", show ty]
20
     nf1 \leftarrow whnf ty1
21
      (x, tyA, tyB) \leftarrow ensurePi nf1
22
      checkType t2 tyA
23
     return (subst x t2 tyB)
   -- ... rest of the cases ommitted for brevity ...
   tcTerm tm Nothing = err ["Must have a type annotation to check ", show tm]
25
26
27
   -- Check-mode
   tcTerm (Lam bnd) (Just ty·(Pi tyA bnd')) = do
29
     tcType tyA
30
      (x, body, _, tyB) \leftarrow Unbound.unbind2Plus bnd bnd'
      extendCtx (TypeSig x tyA) (checkType body tyB)
31
32
     return ty
33 tcTerm (Lam ) (Just nf) = err ["Lambda expression should be a function"]
   -- ... rest of the cases ommitted for brevity ...
   -- if there is no specific case for *checking* the type of the term
   tcTerm tm (Just ty) = do
36
37
     \texttt{ty'} \; \leftarrow \texttt{inferType} \; \texttt{tm}
38
      ty 'equate' ty'
39
      return ty'
```

We can describe the type-checking rules as follows:

- In inference mode:
 - At line 3, we are *inferring* the type of a variable, by consulting it from the type-checking monad's context, according to the I-Var rule in Figure 6.
 - In the function case at line 7, we are *inferring* the type of a dependent function type (pi type), according to the I-Pi rule in Figure 6. We check that the argument type is indeed a type, and then we extend the context with the argument type and check that the body's return type is indeed a type. If both checks are successful, we return the type Type.
 - In the function case at line 12, we are inferring the type of an application, according to the I App Simple rule as stated in Figure 6. We check that the function term is indeed a function type (pi type), and then we check that the provided argument has the same type as the function's argument type. Finally, we return the function's body type, with the argument substituted.
- While in type-checking mode, when we have the information of what type this term needs to have:
 - In the function case at line 28, we are checking the type of a lambda expression to be a dependent function type, according to the C Lambda rule in Figure 6. We check that the function's argument type is indeed a type, and then check that the lambda's body has the same type as the Pi type body return type. If both checks are successful, we return the checked Pi type.
 - In the last case, we are *checking* the type of a term to be equal to a given type, according to the C-Infer-Simple rule in Figure 6. We infer the type of the term, and then check if it is equal to the expected type (using the previously defined *equality among types* function). If both checks are successful, we return the inferred type.

3.5. Examples

The parsing code will be omitted for the sake of brevity (but can be found in the Parser.hs file, in the implementation's repository). We are now going to show some examples of the code in this programming language. First of all we are going to define the Bool, Nat and Vec data types along with Nat's plus function:

```
data Bool : Type where {
  False,
  True
}
data Nat : Type where {
  Zero.
  Succ of (Nat)
plus : (a b : Nat) \rightarrow Nat
plus = \lambda a b. case a of {
  Zero \rightarrow b,
  Succ a' \rightarrow Succ (plus a' b)
}
data Vec (A : Type) (n : Nat) : Type where {
  Nil of [n = 0],
  Cons of (m : Nat) (A) (Vec A m) [n = Succ m]
}
empty_bool_vec : Vec Bool 0
empty_bool_vec = Nil
bool vec : Vec Bool 2
bool_vec = Cons 1 False (Cons 0 True Nil)
```

Next, we are showing how to build a map function that works on vectors (note the type of the vectors having the parameterized length, on the map function type, showing dependent-types being used), we are building a vector and making a proof (static check) of its value:

```
map : (A B : Type) \rightarrow (n : Nat) \rightarrow (f : (A \rightarrow B)) \rightarrow Vec A n \rightarrow Vec B n map = \lambdaA B n f v. case v of {
    Nil \rightarrow Nil,
    Cons n' h t \rightarrow Cons n' (f h) (map A B n' f t)
}

nat_vec : Vec Nat 2
nat_vec = map Bool Nat 2 \lambda(b. case b of {False \rightarrow 10, True \rightarrow 20})
    bool_vec

-- proof that it is a vector like [10, 20]
p_nat_vec : nat_vec = (Cons 1 10 (Cons 0 20 Nil))
```

```
p_nat_vec = refl
```

Another use of dependent-types can be shown with the concat function, showing the returned vector having the length of the sum of the input vectors:

```
concat : (A : Type) \rightarrow (m n: Nat) \rightarrow Vec A m \rightarrow Vec A n \rightarrow Vec A (plus m n) concat = \lambdaA m n a b. case a of { Nil \rightarrow b, Cons m' h t \rightarrow Cons (plus m' n) h (concat A m' n t b) }
```

The aforementioned head function can also be written in our language, along with a proof showing that the returned element is indeed the first element of the vector, and a comment stating a impossible term (the head of an empty vector)):

```
head : (A : Type) → (n : Nat) → Vec A (Succ n) → A
head = λA n v. case v of {
   Cons _ h _ → h
}

p_head : (head Nat 1 nat_vec) = 10
p_head = ref1

-- the following lines do not type-check
-- because when we pass 0 as argument
-- the length of the expected vec argument is (Succ 0), which
-- does not match with the length of the actual vector (which is 0)
-- p_head_empty : (head Bool 0 empty_bool_vec) = False
-- p head empty = ref1
```

3.6. Evaluation

In this section, we presented the implementation of a proof-of-concept programming language with dependent types. We demonstrated the use of dependent types to specify invariants and prove properties about code, and showed how these capabilities can improve code reliability and maintainability. We provided several examples of code written in our language, including functions that manipulate vectors and perform basic operations like mapping and concatenation.

Overall, our implementation illustrates the potential benefits of advanced type systems like dependent types, and demonstrates their ability to enable static checking of invariants and the encoding of logical propositions as types and proofs as code. While our proof-of-concept language is not intended for production use, it serves as a useful case study for exploring the capabilities and potential of dependent types in software development.

3.6.1. Results

In comparing our proof-of-concept language to Agda, a well-established dependently-typed language, it is evident that our language lacks many of the advanced features

and capabilities of Agda. Our language does not support features like *universes*, which allow for the organization and hierarchy of types and their associated terms, and *implicit parameters*, which allow for the implicit passing of arguments in function definitions. These features are present in Agda and can be useful for a wide range of applications. It is clear that our language is limited in its capabilities and may not be suitable for all types of projects that may benefit from the use of dependent types.

4. Conclusion

In this thesis, we aimed to examine the potential advantages of utilizing advanced type systems, such as dependent types, in software development through the implementation of a dependently-typed programming language. Our specific objectives included investigating existing work on dependently-typed languages, developing a proof-of-concept language, using the language to encode business invariants and demonstrate the ability to prove properties about the code.

To support these objectives, we presented a dependently-typed programming language with a type-checker and a parser, written in Haskell. The language is based on the lambda calculus and has a simple syntax, with a type system that supports dependent types. While the language is not intended for production use, it serves as a proof-of-concept and a starting point for learning about dependently-typed programming languages.

Through the implementation and analysis of this language, we were able to gain a deeper understanding of how advanced type systems such as dependent types work and their potential benefits for improving the reliability and safety of software. However, it is important to note that the use of dependent types also comes with trade-offs and limitations, including the added burden of having to prove properties about the code and the potential for increased complexity in the codebase and development process. While the benefits of dependent types may outweigh the costs in certain cases, such as for critical or safety-sensitive systems, it is important to carefully evaluate the trade-offs and limitations for each specific application.

Overall, our findings suggest that dependent types have the potential to significantly improve the reliability and safety of software, but more research and development is needed to fully realize this potential in practice. The language and related resources can be found at https://github.com/eduhenke/dep-tt.

References

[Church 1941] Church, A. (1941). The calculi of lambda-conversion.

[Mörtberg and Norell 2022] Mörtberg, A. and Norell, U. (2022). Agda. https://agda.readthedocs.io/en/v2.6.2.2/. [Online; accessed 25-October-2022].

[nLab authors 2022a] nLab authors (2022a). equality. http://ncatlab.org/nlab/show/equality. Revision 27.

[nLab authors 2022b] nLab authors (2022b). judgment. http://ncatlab.org/nlab/show/judgment. Revision 22.

- [nLab authors 2022c] nLab authors (2022c). proofs as programs. https://ncatlab.org/nlab/show/proofs+as+programs. Revision 5.
- [nLab authors 2022d] nLab authors (2022d). propositions as types. https://ncatlab.org/nlab/show/propositions+as+types. Revision 41.
- [Pierce 2002] Pierce, B. C. (2002). Types and Programming Languages. The MIT Press, 1st edition.
- [Pierce 2004] Pierce, B. C. (2004). Advanced Topics in Types and Programming Languages. The MIT Press.
- [Team 2022a] Team, T. C. D. (2022a). Coq. https://coq.inria.fr/. [Online; accessed 25-October-2022].
- [Team 2022b] Team, T. I. D. (2022b). Idris. https://www.idris-lang.org/. [Online; accessed 25-October-2022].
- [Team 2022c] Team, T. L. D. (2022c). Lean. https://leanprover.github.io/. [Online; accessed 25-October-2022].
- [Weirich 2022] Weirich, S. (2022). Implementing Dependent Types in pi-forall. https://github.com/sweirich/pi-forall/blob/2022/doc/oplss.pdf. [Online; accessed 25-July-2022].